

Contents

1 Address resolution protocol

The *address resolution protocol* or **ARP** is a standalone module which performs the mapping of a given layer-3 address $addr_{NL}$ to another address, the link-layer address, which we will call $addr_{LL}$.

1.1 Why do we need this?

The reason that we require this $addr_{LL}$ is because when we need to send data to a host we do so over a link which is indicated in the routing table for said packet.

However, links don't speak the same network-layer protocol as twine - they speak whatever protocol they implement - i.e. Ethernet via **LIInterface** or the in-memory **PipedLink**. Needless to say there is also always a requirement of such a mapping mechanism because several links may be backed by a different link-layer protocols in their **Link** implementation and therefore we cannot marry ourselves to only one link-layer protocol - *we need something dynamic*.

1.2 The mapping function

We now head over to the technical side of things. Before we jump directly into an analysis of the source code it is worth considering what this procedure means in a mathematical sense because at a high-level this is what the code is modeled on.

If we have a router r_1 which has a set of links $L = \{l_1, l_2\}$ and we wanted to send a packet to a host addresses h_1 and h_2 which are accessible over l_1 and l_2 respectively then the mapping function would appear as such:

$$(h_1, l_1) \rightarrow addr_{LL_1}$$

$$(h_2, l_2) \rightarrow addr_{LL_2}$$

On the right hand side the $addr_{LL_1}$ and $addr_{LL_2}$ are the resolved link-layer addresses.

$$(h_i, l_i) \rightarrow addr_{LL_i}$$

Therefore we discover that we have the above mapping function which requires the network-layer h_i address you wish to resolve and the link l_i over which the resolution must be done, this then mapping to a single scalar - the link-layer address, $addr_{LL_i}$.

1.3 Implementation

We will begin the examination of the code at the deepest level which models this mathematical function earlier, first, after which we will then consider the code which calls it and how that works.

1.3.1 The entry type

Firstly let us begin with the definition of the in-memory data type which holds the mapping details. this is known as the `ArpEntry` struct and it is shown in part below:

```
1 public struct ArpEntry
2 {
3     private string l3Addr;
4     private string l2Addr;
5
6     ...
7
8     public bool isEmpty()
9     {
10         return this.l3Addr == "" && this.l2Addr == "";
11     }
12
13     public static ArpEntry empty()
14     {
15         return ArpEntry("", "");
16     }
```

Please note the methods `isEmpty()`. An entry is considered empty if both its network-layer and link-layer fields have an empty string in them, this is normally accomplished by calling the `empty()` static method in order to construct such an `ArpEntry`.

1.3.2 Making an ARP request

The code to make an ARP request is in the `regen(Target target)` method and we will now go through it line by line.

1.3.2.1 Setting up the request and link Firstly we are provided with a `Target`, this encapsulates the network-layer address and the `Link` instance we want to request over. We now extract both of those items into their own variables:

```
1 // use this link
2 Link link = target.getLink();
3
```

```

4 // address we want to resolve
5 string addr = target.getAddr();

```

Before we make the request we will need a way to receive the response, therefore we attach ourselves, the `ArpManager`, as a `Receiver` to the link:

```

1 // attach as a receiver to this link
2 link.attachReceiver(this);
3
4 logger.dbg("attach done");

```

This provides us with a callback method which will be called by the `Link` whenever it receives *any* traffic. It is worth noting that such a method will not run on the thread concerning the code we are looking at now but rather on the thread of the `Link`'s - we will discuss later how will filter it and deliver the result to us, *but for now - back to the code*.

1.3.2.2 Encoding and sending the request Now that we know what we want to request and over which link we can go ahead and encode the ARP request message and broadcast it over the link:

```

1 // generate the message and send request
2 Arp arpReq = Arp.newRequest(addr);
3 Message msg;
4 if(toMessage(arpReq, msg))
5 {
6     link.broadcast(msg.encode());
7     logger.dbg("arp req sent");
8 }
9 else
10 {
11     logger.error("Arp failed but oh boy, at the encoding level");
12 }

```

As you can see we make use of the `broadcast(byte[])` method, this is handled by the link's implementation according to its link-layer protocol.

1.3.2.3 Waiting for a response We now have to wait for a response and not just any response. It has to be an ARP reply for the particular network-layer address we requested earlier.

This is done with the following code:

```

1 // wait for reply
2 string llAddr;
3 bool status = waitForLLAddr(addr, llAddr);
4
5 ...

```

As you can see we have this call to a method called `waitForLLAddr(addr, llAddr)`. This method will block for us and can wake up if it is signaled to by the callback method running on the `Link`'s thread (as mentioned previously).

```
1  Stopwatch timer = Stopwatch(AutoStart.yes);
2
3  // todo, make timeout-able (todo, make configurable)
4  while(timer.peek() < this.timeout)
5  {
6      this.waitLock.lock();
7
8      scope(exit)
9      {
10         this.waitLock.unlock();
11     }
12
13     this.waitSig.wait(dur!("msecs")(500)); // todo, duty cycle if missed notify but also hel
14
15     // scan if we have it
16     string* llAddr = l3Addr in this.addrIncome;
17     if(llAddr != null)
18     {
19         string llAddrRet = *llAddr;
20         this.addrIncome.remove(l3Addr);
21         llAddrOut = llAddrRet; // set result
22         return true; // did not timeout
23     }
24 }
25
26 return false; // timed out
```

Because it is implemented using a condition variable, it could potentially miss a signal from the calling `notify()` if we only call `wait()` on our thread *after* the link's thread has called `notify()`. Therefore, we make our `wait()` wake up every now and then by using a timed-wait, to check if the data has been filled in by the other thread.

Second of all, what we do after retrying from `wait(Duration)` is check if the *requested network-layer address* has been resolved or not - this is that filtering I was mentioning earlier. This is important as we don't want to wake up for *any* ARP response, but only the one which matches our `addr` requested.

Thirdly, this also gives us a chance to check the while-loop's condition so that we can see if we have timed out (waited long enough) for an ARP response.

After all is done, the resulting entry is placed in a globally accessible `string[string] addrIncome` which is protected by the `waitLock` for both threads contending it. We then continue:

```
1  ...
2
3  // if timed out
4  if(!status)
5  {
6      logger.warn("Arp failed for target: ", target);
7      return ArpEntry.empty();
8  }
9  // on success
10 else
11 {
12     ArpEntry arpEntry = ArpEntry(addr, llAddr);
13     logger.info("Arp request completed: ", arpEntry);
14     return arpEntry;
15 }
```

We now check, as I said, if the entry is valid or not. If we timed-out then we would have returned `false`. Now, as we shall see later, we will still have to return *some* `ArpEntry` because that is the signature of our method, `regen(Target target)`. Thus, if we failed to get an `ArpEntry` we then return one generated by `ArpEntry.empty()`, else we return the actual entry that we received.

1.3.2.4 Catching responses I have mentioned that the thread which waits for a matching ARP response to come in (the one which calls the `wait(Duration)` above. So then, the question is - which thread is the one calling `notify()` on the condition variable and under which scenarios?

Recall that we attached the `ArpManager` as a `Receiver` to the `Link` object which was passed into the `regen(Target)` method:

```
// use this link
Link link = target.getLink();

// address we want to resolve
string addr = target.getAddr();

// attach as a receiver to this link
link.attachReceiver(this);

logger.dbg("attach done");
```

Now the reason for this is that whenever traffic is received on a **Link** it will copy the **byte[]** containing the payload to each attached **Receiver**.

This means that the **ArpManager** will receive all packets from a given link, the question is - which ones to we react to? Well that's easy. Below I show you the **onReceive(Link src, byte[] data, string srcAddr)** method which the arp manager overrides. This is called every time a given link receives data:

```
1  /**
2   * Called by the `Link` which received a packet which
3   * may be of interest to us
4   *
5   * Params:
6   *   src = the `Link` from where the packet came from
7   *   data = the packet's data
8   *   srcAddr = the link-layer source address
9   */
10 public override void onReceive(Link src, byte[] data, string srcAddr)
11 {
12     Message recvMesg;
13     if(Message.decode(data, recvMesg))
14     {
15         // payload type
16         if(recvMesg.getType() == MType.ARP)
17         {
18             Arp arpMesg;
19             if(recvMesg.decodeAs(arpMesg))
20             {
21                 logger.dbg("arpMesg, received: ", arpMesg, "from: ", srcAddr);
22                 ArpReply reply;
23                 if(arpMesg.getReply(reply))
24                 {
25                     logger.info("ArpReply: ", reply);
26
27                     // place and wakeup waiters
28                     placeLLAddr(reply.networkAddr(), reply.llAddr());
29                 }
30
31                 ...
32                 ...
33                 ...
34                 ...
35     }
```

What we do here is we attempt to decode each incoming packet into our **Message** type, then further check if it is an ARP-typed message. If this is the case then

we check if it is an ARP request (because as we have seen, ARP requests are **not** handled here).

```
1  /**
2   * Called by the thread which has an ARP response
3   * it would like to pass off to the thread waiting
4   * on the condition variable
5   *
6   * Params:
7   *   l3Addr = the network layer address
8   *   llAddr = the link-layer address
9   */
10 private void placeLLAddr(string l3Addr, string llAddr)
11 {
12     this.waitLock.lock();
13
14     scope(exit)
15     {
16         this.waitLock.unlock();
17     }
18
19     this.waitSig.notify(); // todo, more than one or never?
20
21     this.addrIncome[l3Addr] = llAddr;
22 }
```

If this is the case then we will place the link-layer address into a key-value map where the key is the network-layer address and the value is the link-layer address. After this we wake up the sleeping thread by calling `notify()`.

1.3.3 Caching

I mentioned that there is caching involved. The involvement is that all `ArpEntry`'s are stored in a `CacheMap! (ArpEntry)` which means that they will exist in there for some period of time and then be evicted.

If an entry has not yet been cached-in then it is created on demand when you do `map.get(Target)`. Now remember the `regen(Target)` method? Well, that's the regeneration method that we supply this cache map upon instantiation - therefore it works as expected.

1.4 The API

We have now discussed the gritty internals which aid us in creating requests, awaiting replies and then returning the matched entry. We now must move over to the publicly facing API of the `ArpManager`. This really just contains a single method:

```
Optional!(ArpEntry) resolve(string networkAddr, Link onLink)
```

The way this method works is that it will return an `Optional!(ArpEntry)`, meaning that you can test to see if the arp resolution process succeeded or failed (i.e. timed-out for example) using code that looks akin to what shall follow.

I have prepared an example which can illustrate the usage of the `ArpManager`. In fact this example is part of a unittest which tests the various scenarios that can occur with the manager itself.

1.4.1 Mock links

Firstly we setup a pseudo-link. This is a sub-class of the `Link` class which is specifically configured to respond **only** to ARP requests and only to those which a mapping exists for.

In this example I configure two mappings of network-layer addresses to link-layer addresses:

$$(host_{A_{13}}, dummyLink) \rightarrow host_{A_{12}}$$

$$(host_{B_{13}}, dummyLink) \rightarrow host_{B_{12}}$$

The code to do this is as follows:

```
1 // Map some layer 3 -> layer 2 addresses
2 string[string] mappings;
3 mappings["hostA:13"] = "hostA:12";
4 mappings["hostB:13"] = "hostB:12";
5
6 // create a dummy link that responds with those mappings
7 ArpRespondingLink dummyLink = new ArpRespondingLink(mappings);
```

1.4.2 Resolution

We then must create an `ArpManager` we can use for the resolution process:

```
ArpManager man = new ArpManager();
```

Now we are ready to attempt resolution. I first try to resolve the link-layer address of the network-layer address `hostA:13` by specifying it along with the mock link, `dummyLink`, which we created earlier:

```
1 // try resolve address `hostA:13` over the `dummyLink` link (should PASS)
2 Optional!(ArpEntry) entry = man.resolve("hostA:13", dummyLink);
3 assert(entry.isPresent());
4 assert(entry.get().llAddr() == mappings["hostA:13"]);
```


In the above case the mapping succeeds and we get an `ArpEntry` returned from `entry.get()`, upon which I extract the link-layer address by calling `llAddr()` on it and comparing it to what I expected, `mappings["hostA:13"]` - which maps to `hostA:12`.

We do a similar example for the other host:

```
1 // try resolve address `hostB:13` over the `dummyLink` link (should PASS)
2 entry = man.resolve("hostB:13", dummyLink);
3 assert(entry.isPresent());
4 assert(entry.get().llAddr() == mappings["hostB:13"]);
```

Lastly, I wanted to show what a failure would look like. With this we expect that `entry.isPresent()` would return `false` and therefore stop right there:

```
1 // try top resolve `hostC:13` over the `dummyLink` link (should FAIL)
2 entry = man.resolve("hostC:13", dummyLink);
3 assert(entry.isPresent() == false);
```

This resolution fails because our `ArpRespondingLink`, our *dummy link*, doesn't respond to mapping requests of the kind $(host_{B_{13}}, dummyLink)$.

1.4.3 Shutting it down

We need to shut down the `ArpManager` when we shut down the whole system, this is then accomplished by running its destructor:

```
// shut down the arp manager
destroy(man);
```

2 Routing and forwarding

Routing is the process by which one announces their routes to others, whilst forwarding is the usage of those learnt routes in order to facilitate the transfer of packets from one endpoint to another through a network of inter-connected routers.

2.1 A route

Before we can get into the process of routing we must first have a conception of a *route* itself.

A route consists of the following items:

1. A *destination*
 - Describes to whom this route is for, i.e. a route to *who*

2. A *link*
 - The `Link` object over which we can reach this host
3. A *gateway*
 - This is who we would need to forward the packet *via* in order to get the packet either to the final destination (in such a case the *gateway* = *destination*) or the next-hop gateway that we must forward via (*gateway* \neq *destination*)
4. A *distance*
 - This is metric which doesn't affect *how* packets are forwarded but rather how routes that have the same matching *destination* are tie-broken.
 - Given routes $r = \{r_1, r_2\}$ and a function $d(r_i)$ which returns the distance we shall install the route r_i which has the lowest distance, hence $r_{installed} = r_i \text{ where } d(r_i) = \min(d(r))$ (TODO: fix this maths)
5. A *timer* and *lifetime*
 - We have *timer* which ticks upwards and a *lifetime* which allows us to check when the *timer* > *lifetime* which signifies that the route has expired

And in code this can be found as the `Route` struct shown below:

```

1  /**
2   * Represents a route
3   */
4  public struct Route
5  {
6      private string dstKey; // destination
7      private Link ll; // link to use
8      private string viaKey; // gateway (can be empty)
9      private ubyte dst; // distance
10
11     private Stopwatch lifetime; // timer
12     private Duration expirationTime; // maximum lifetime
13
14     ...
15 }

1
2  /**
3   * Constructs a route to a destination
4   * over a given link with a given metric.
5   *
6   * The destination of this route is
7   * directly reachable over the link.
8   *
9   * Params:
10    *   dst = the destination network-layer

```

```

11     * address
12     *   link = the `Link`
13     *   distance = the distance
14     */
15     this(string dst, Link link, ubyte distance)
16     {
17         this(dst, link, dst, distance);
18     }
19
20     /**
21     * Constructs a route to a destination
22     * over a link with a given metric.
23     *
24     * This also let's you set the next-hop
25     * gateway that should be used.
26     *
27     * Params:
28     *   dst = the destination network-layer
29     *   address
30     *   link = the `Link`
31     *   via = the next-hop gateway's address
32     *   distance = the distance
33     */
34     this(string dst, Link link, string via, ubyte distance, Duration expirationTime = dur!(0))
35     {
36         this.dstKey = dst;
37         this.ll = link;
38         this.viaKey = via;
39         this.dst = distance;
40
41         this.lifetime = Stopwatch(AutoStart.yes);
42         this.expirationTime = expirationTime;
43     }
44
45     /**
46     * Is this route direct? As
47     * in the destination is
48     * directly reachable via
49     * the gateway (i.e. the
50     * destination matches the
51     * gateway)
52     *
53     * Returns: `true` if so,
54     * otherwise `false`
55     */

```

```

56     public bool isDirect()
57     {
58         return this.dstKey == this.viaKey; // todo, should we ever use cmp?
59     }
60
61     /**
62      * Checks if this route
63      * has expired
64      *
65      * Returns: `true` if so,
66      * `false` otherwise
67      */
68     public bool hasExpired()
69     {
70         return this.lifetime.peek() > this.expirationTime;
71     }
72
73     /**
74      * Resets the expiration
75      * timer for this route
76      */
77     public void refresh()
78     {
79         this.lifetime.reset();
80     }
81
82     /**
83      * Retrieves this route's
84      * destination address
85      *
86      * Returns: the address
87      */
88     public string destination()
89     {
90         return this.dstKey;
91     }
92
93     /**
94      * Retrieves this route's
95      * associated link
96      *
97      * Returns: a `Link`, or
98      * `null` if this is a
99      * self-route
100     */

```

```

101     public Link link()
102     {
103         return this.ll;
104     }
105
106     /**
107      * Returns whether or not
108      * this route is a self-route
109      * (i.e. the link set was
110      * `null`)
111      *
112      * Returns: `true` if so,
113      * otherwise `false`
114      */
115     public bool isSelfRoute()
116     {
117         return this.ll is null;
118     }
119
120     /**
121      * Retrieves the gateway
122      * of this route
123      *
124      * Returns: the gateway's
125      * address
126      */
127     public string gateway()
128     {
129         return this.viaKey;
130     }
131
132     /**
133      * Retrieves the distance
134      *
135      * Returns: the distance
136      * metric
137      */
138     public ubyte distance()
139     {
140         return this.dst;
141     }
142
143     /**
144      * Compares two routes with one
145      * another

```

```

146     *
147     * Params:
148     *   r1 = first route
149     *   r2 = second route
150     * Returns: `true` if the routes
151     * match exactly, otherwise `false`
152     */
153     public static bool isSameRoute(Route r1, Route r2)
154     {
155
156         return r1.destination() == r2.destination() &&
157             r1.gateway() == r2.gateway() &&
158             r1.distance() == r2.distance() &&
159             r1.link() == r2.link();
160     }
161
162     /**
163     * Compares this `Route` with
164     * another
165     *
166     * Params:
167     *   rhs = the other route
168     * Returns: `true` if the routes
169     * are identical, `false` otherwise
170     */
171     public bool opEquals(Route rhs)
172     {
173         return isSameRoute(this, rhs);
174     }
175 }

```